
News  

US Training Days 
Our US training days are now confirmed for the 15th and 16th of October at the Embassy 
Suites Chicago Downtown hotel, prior to the Society for Neuroscience annual meeting. 
The training days are offered at $250 per day, including continental breakfast buffet and 
lunch as well as coffee and tea, and are suitable for both existing and prospective users 
of Spike2 and Signal. They include introductions to the software along with sessions on 
script writing, stimulus generation and on-line control. Our popular workshop thread, 
where you can discuss your questions and requirements with a CED engineer or 
programmer in a small group, will also be available. 

If you or any of your colleagues are interested in attending, please follow the links to view 
the training day program and complete the registration form. 

Meetings and events 
Scandinavian Physiological Society 2009 
Uppsala Biomedicinska Centrum 
Uppsala, Sweden 
August 14th – 16th 2009 

 

The 6th Congress of the International Society for Autonomic 
Neuroscience 
ISAN09 is a joint meeting to be held with the European Federation of 
Autonomic Societies (EFAS) in Sydney, from September 1st-4th, 
2009. Approx. 400 international and national delegates are expected, 
having clinical and/or basic scientific research interests in autonomic 
neuroscience. 

Plenary speakers: Darwin Berg (UCSD), Hugo Critchley (University of Sussex), Max Hilz 
(University of Erlangen-Nuremberg) and Elspeth McLachlan (POWRI) 

Further details of the meeting can be found at http://www.iceaustralia.com/isan2009/ 

Proudly supported by Cambridge Electronic Design and the Foundation for High Blood 
Pressure Research, Australia. 

 

9th International Congress of The Polish Neuroscience Society 
University of Warsaw 
Warsaw, Poland 
September 9th – 12th 2009 

Southampton Neuroscience Group (SoNG) Annual Meeting 
University of Southampton 
Southampton, UK 
September 25th 2009 

Neuroscience 2009 - 39th Annual Meeting 
McCormick Place 
Chicago, IL, USA 
October 17th – 21st 2009 

Latest versions of Spike2 and Signal 
Updates for Spike2 and Signal are available from the CED downloads page, or by 
clicking on the links in the table below. Demonstration versions of the latest software are 
also available.   

 Spike2 downloads Signal downloads 
Spike2 version 7.01 Signal version 4.05 

Spike2 version 6.12 Signal version 3.12 

Spike2 demo Signal demo 

Spike2   
Q. I am using E-prime software to generate various stimuli for my psychological 

experiments and would like to interface this with Spike2 so that the stimulus type 
is marked in the data file. Is there any way to log markers from E-prime in Spike2?  

A. The 1401 family has 16 digital inputs and 16 digital outputs. Digital input bits 15-8 
are the event ports and Digital input bits 7-0 are used for the Digital marker 
channel as shown by the diagram below. 

 
Experiment design programs such as E-prime and Presentation allow the user to 
generate video and image files as a stimulus and output digital marker 
information relating to the current stimulus on a 25 way D-type parallel port. This 
can then be connected to the 25 way digital input connector on the 1401 using an 
appropriately wired cable. 

 
Wiring diagram from parallel port (left) to 1401 digital input (right) 

 

Back to top

 

 

http://www.ced.co.uk/uplsu.shtml
http://www.ced.co.uk/s2wu7u.shtml
http://www.ced.co.uk/s2wu6u.shtml
http://www.ced.co.uk/s2demu.shtml
http://www.ced.co.uk/sigw4u.shtml
http://www.ced.co.uk/sigw3u.shtml
http://www.ced.co.uk/sigdemu.shtml


To record the E-prime or Presentation marker data to the Spike2 file, set-up a 
digital marker channel (channel 32) in the sampling configuration. Codes sent by 
the program will then be displayed in the marker channel during sampling. 

 

  

 

Q. I would like to be able to take data from multiple average result views and display 
them overdrawn in a single view. 

A. It is possible to add ‘blank’ frames onto the end of Signal data files and memory 
views by using the Append frame command from the Analysis menu. Data can 
then be copied from any file that includes the same number of channels and 
same sample rate using the standard Ctrl+C keyboard shortcut, and pasted into 
the new ‘blank’ frame by clicking on the view and using the Ctrl+V shortcut (or 
the Edit menu Copy and Paste commands). Once you have added the frames 
from all average result files into the original result view simply select the frames 
you want to include in the overdraw using the Overdraw settings dialog and then 
use the Overdraw frames command (both available from the View menu) to apply 
these to the current file.  

  
Overdraw settings dialog 

Scripts: Spike2  
 Spike2 version 7.01 includes new script commands that give access to the 

mouse positions and mouse button clicks in data files, result and XY views when 
using a script toolbar or user-defined dialog. The attached script, 
MouseEvent.s2s, is an example that allows the user to add TextMark items to 
a data region interactively by holding Shift and clicking and dragging the 
mouse pointer. 

Scripts: Signal  
Q. I use Signal to give subject feedback during experiments by generating a trend 

plot over multiple frames based on the response. What I would like to do is 
display this trend plot at the end of a frame for a short period and then hide it 
from view so the subject only has a brief time to see their progress. Is this 
possible using a script? 

A. The attached script, FlashTrendPlot.sgs, is an example that pauses 
sampling at the end of each sweep and displays an associated trend plot for a 
user-defined length of time, before hiding it and continuing with sampling. This 
example trend plot is set to take a peak to peak measurement between cursor 1 

CED
Digital markers can be displayed as vertical markers to span across all channels in Spike2 v7

CED
The 3D drawing options can be used to give a waterfall-type display of all overdrawn frames


'$MouseEvent|Example script to show interaction of mouse position and button clicks with a script
'The script allows the user to interactively define areas of data with textmarks drawn in states mode

'The script is a 'work in progress' and is offered without guarantees.
'You must test it to see whether it is suitable for your application.

'CED 19/07/09

var data%;      'View handle of data file
var mCh%;       'Memory channel for TextMarks
var code%[4];   'Code for TextMark
var pos;

if App(-1)<701 then Message("This script needs at least Spike2 version 7.01"); halt endif;	'Checks for the latest version of Spike2


'Define a new mouse pointer
var mc% := MousePointer(
"             i|"
"              |"
"             i|"
"              |"
"             i|"
"              |"
"             i|"
"              |"
"        iiiiiiiiiii|"
"        ii       ii|"
"        i i     i i|"
"        i  i   i  i|"
"        i         i|"
"i i i i i    h     i i i i i|"
"        i         i|"
"        i  i   i  i|"
"        i i     i i|"
"        ii       ii|"
"        iiiiiiiiiii|"
"             |"
"            i|"
"             |"
"            i|"
"             |"
"            i|"
"             |"
"            i|");

ToolbarSet(1, "Quit");
ToolbarSet(2, "Open file", Open%);
ToolbarSet(4, "Add event", SetAdd%);
ToolbarSet(5, "Delete event", SetDel%);
ToolbarSet(6, "Delete all", DelAll%);
ToolbarEnable(4,0);
ToolbarEnable(5,0);
ToolbarEnable(6,0);
Toolbar("Interactive event editing - Press Shift and click and drag to add time range",1023);


Func Open%()
data% := FileOpen("",0,2);      'Prompt user to open a file
if data% > 0 then           'If file opened correctly
    mCh% := MemChan(8,100);     'Create a new memory channel for TextMarks
    DrawMode(mCh%,15);      'Draw in States mode
    ChanShow(mCh%);             'Show channel
    VerticalMark(mCh%,1);       'Set-up vertical marker display
    Window(0,0,100,100);        'Show data file full screen
    FrontView(data%);
    SetAdd%();
endif;
return 1;
end;


Func SetAdd%()
ToolbarText("Interactive event editing - Press Shift and click and drag to add time range");
ToolbarMouse(data%,-1,4,4,AddEvent%,AddUp%,Move%);  'Access mouse position
ToolbarEnable(4,0);
ToolbarEnable(5,1);
ToolbarEnable(6,1);
return 1;
end;


Func SetDel%();
ToolbarText("Click on a vertical marker to delete");
ToolbarMouse(data%,-1,2,0,DelEvent%,DelUp%,Move%);
ToolbarEnable(4,1);
ToolbarEnable(5,0);
return 1;
end;

Func DelAll%();
MemDeleteTime(mCh%,3,0.0,MaxTime());
SetAdd%();
return 1;
end;


Func AddEvent%(vh%,ch%,x,y,flags%)  'When left button is held down
var say$;
code%[0] := 1;      'Marker code 01
MemSetItem(mCh%,0,x,code%[]);       'Create a TextMark
pos:=x;
return mc% + 512;
end;

Func DelEvent%(vh%,ch%,x,y,flags%)
MemDeleteTime(mCh%,0,x,1);      'Delete TextMark
MemDeleteTime(mCh%,0,Nexttime(mCh%,x));
return mc%;
end;

Func AddUp%(vh%,ch%,x,y,flags%);    'When left button is released
code%[0] := 0;
if pos <> x then
    MemSetItem(mCh%,0,x,code%[]);   'Set TextMark code 00
else;
    MemDeleteTime(mCh%,0,x,1);      'Delete TextMark
    endif;
return 1;
end;

Func DelUp%(vh%,ch%,x,y,flags%);       'This returns the state of the left mouse button 

return 1;
end;

Func Move%(vh%,ch%,x,y,flags%);         

return mc%;
end;

CED
MouseEvent.s2s


'$FlashTrendPlot| Displays an associated trend plot for a time set by the user at the end of the sampled frame

'The script is a 'work in progress' and is offered without guarantees.
'You must test it to see whether it is suitable for your application.

'CED 16/06/09

var fTime:= 0.02;						'Display time for trend plot (s)
var data%;								'Data file being sampled
var tPlot%;
var ch%;
var ok%;
var flag%;

HideAll();
'Set up toolbar
'View(SampleHandle(1)).WindowVisible(1);
ToolbarSet(1, "&Quit", Quit%);				
ToolbarSet(3, "&New file", NewFile%);
ToolbarSet(4, "&Start Sampling", Start%);
ToolbarEnable(4,0);
Toolbar("'Flash' display of trend plot. Position Cursors for measurements and click Start Sampling",1023);


'Open new file for sampling
Func NewFile%()
data% := FileNew(0);		'Open new file based on current sampling configuration
if data% > 0 then			'If file opened successfully
	Window(0,46,91,95);
    ChanShow(-1,1);
    FrontView(data%);		'Make sure data is current view
    DlgCreate("Channel settings");
    DlgChan(1,"Channel to Measure",1);
    DlgReal(2,"Time to display plot (s)",0.01,1);
    ok%:= DlgShow(ch%,fTime);
    if ok% < 0 then halt;endif;
    XRange(MinTime(),MaxTime());	'Scale X-axis
	ToolbarEnable(4,1);		'Enable Start button
	ToolbarEnable(3,0);		'Disable New button
	SamplePause(1);			'Set pause at end of sweep
    CursorSet(2);
    View(SampleHandle(1)).WindowVisible(1);

    'Set-up trend plot
    tPlot%:=MeasureToXY(0);
    Window(0,0,91,46);
    MeasureX(104,ch%,"MaxTime()","0",0);
    MeasureY(9,ch%,"Cursor(2)","Cursor(1)",0);
    MeasureChan(1,"Channel 1");
    ProcessOnLine(-1,-1,0,0,1,0,0);
    ViewColour(0,4);   'Set view item colour
    XYJoin(-1,0);   'Set XY join mode
    YRange(-1,0,5);
    
else
	Message("Failed to create new data file\nCheck 1401 is connected and retry");
endif;
return 1;
end;

Func Quit%()
SampleStop();		'Stop sampling
View(tPlot%).WindowVisible(1);  'Show trend plot
RestoreAll();
return 0;
end;

'Start sampling
Func Start%()
View(data%);
SampleStart();		'Start sampling
ToolbarSet(0,"Idle",Idle%);		'Set Idle routine to monitor data
ToolbarClear(4);				
ToolbarSet(4, "&Stop", Stop%);	'Set Stop button
return 1;
end;


'Stop sampling
Func Stop%()
View(data%);
SampleStop();		'Stop sampling
ToolbarClear(4);	
ToolbarSet(4, "Start", Start%);		'Set Start button
ToolbarEnable(4,0);		'Disable Start button	
ToolbarEnable(3,1);		'Enable New file button
ToolbarClear(0);
View(tPlot%).WindowVisible(1);  'Show trend plot
return 1;			'Clear Idle routine
end;


'Idle routine to monitor when sweep is taking place
Func Idle%()									
View(data%);

	if SampleStatus() = 3 then				'If paused at end of sweep
        FrontView(tPlot%);                      'Show plot
        Yield(fTime);                           'Hold display for set time
        View(tPlot%).WindowVisible(0);          'Hide plot
        SampleSweep();						'Release pause to prepare for next sweep
        FrontView(data%);
	endif;
return 1;
end;

var gFloat%[20];                'global for floating window states
proc HideAll()
var i%;
gFloat%[0] := App(-2);          'number of windows
for i% := 1 to gFloat%[0] do    'hide all windows and save state
   gFloat%[i%] := View(App(i%)).WindowVisible(0);
   next;
end

proc RestoreAll()
var i%;
for i% := 1 to gFloat%[0] do    'restore hidden windows
   View(App(i%)).WindowVisible(gFloat%[i%]);
   next;
end



CED
FlashTrendPlot.sgs



and cursor 2 on the selected channel, but could easily be modified to give other 
measurement options. 

Did you know…? 
You can adjust the sampling rate of a data channel off-line using the Channel Process 
Interpolate or Match channel commands. This is useful if you want to create a multi-
channel result view, as all included channels need to be sampled at the same rate. 

Recent questions   
Q. Is there any way to add text, such as stimulus type or current experiment 

conditions, to a Spike2 data file during sampling? 

A. The TextMark data channel allows the user to insert timed comments into a data 
file while sampling. Simply select TextMark from the Type drop-down list when 
adding a new channel in the sampling configuration to enable channel 30 as a 
TextMark channel.  

When sampling, use the Ctrl+T keyboard shortcut to open the Create TextMark 
dialog. There are two modes for adding TextMark data to the file: clicking the 
Save button adds a TextMark to the file at the current time. You can edit the text 
and code of a TextMark at anytime by double-clicking on the TextMark in channel 
30. The Set Time button records the time of the button press and allows the user 
to set the text string and code. Clicking the Save button after Set Time writes the 
TextMark at the recorded time.  

 
TextMark data drawn in States mode with Vertical Markers 

TextMark data can be drawn in state mode and also displayed as vertical 
markers to show the included text, as above. 

User group      
Try the new CED Forums bulletin board for software and hardware support 

If you have any comments about the newsletter format and content, or wish to un-
subscribe from the mailing list for this newsletter, please notify sales@ced.co.uk. 
All Trademarks are acknowledged to be the Trademarks of the registered holders. 

CED
The code used determines the colour of the TextMark. The colours used are the same as those used for WaveMark data, which can be edited in the Spike2 Colour palette

